

INSTITUTO DE ESTUDOS GEOGRÁFICOS
FACULDADE DE LETRAS — UNIVERSIDADE DE COIMBRA



Cadernos de Geografia

SOFTWARE
PARA A MEDIDA DE CONCENTRAÇÃO ESPACIAL
 $C(N,r)$

GABRIELA SALGUEIRO *
ARMANDO BERNARDO LACERDA **

O estudo da concentração espacial de uma distribuição de pontos por meio da medida

$$C(N,r) = (N - n(r)) / (N - 1)$$

onde N é o número de pontos e $n(r)$ o número de manchas para um raio r , (vide KOSTRUBIEC e SALGUEIRO, 1986), pode ser muito facilitado, e mesmo enriquecido, pelo uso de adequado apoio informático.

É sabido que as manchas assumem diferentes configurações, em função dos incrementos do raio r com que são desenhados os N círculos, nos N pontos ($1 < N < +\infty$) de cada distribuição pontual. A possibilidade de rápida visualização dessas sucessivas configurações, pelo simples conhecimento das coordenadas de localização dos pontos relativamente a um sistema de eixos de referência, era já uma grande vantagem, conseguida com o programa utilizado no caso particular do ensaio apresentado em 1986. Juntava-se-lhe ainda a construção dos grafos, que muito auxiliava a contagem, $n(r)$, das manchas, para cada valor do raio.

Não sendo fácil, com os programas utilizados no citado ensaio, a adaptação a outros tipos de escalas, surgiu a necessidade de implementar software que, não só se mostrasse receptivo a quaisquer distribuições espaciais de pontos (a duas dimensões), como permitisse até compará-las. Esses objectivos são já atingidos com o programa que os autores vêm propor e, que junta aos já citados benefícios algumas novas possibilidades, para além de proporcionar uma cómoda utilização.

* Instituto de Estudos Geográficos, Faculdade de Letras de Coimbra.

** Centro Regional de Viseu da Universidade Católica Portuguesa.

APRESENTAÇÃO SUMÁRIA DO PROGRAMA

1. Quanto à linguagem:

A presente versão, vem expressa em *GW-BASIC Rev. 3.12* da *Olivetti Personal Computer*, que corre sobre o sistema operativo *MS-DOS 3.2*.¹

2. Criação de ficheiros:

Cada distribuição pontual é introduzida no programa sob a forma de um ficheiro, onde são guardados o número de pontos e as suas coordenadas, bem como a escala da carta onde se encontram representados. Os valores entrados e mostrados no écran são susceptíveis de posterior correcção.

3. Gestão conjunta de vários ficheiros:

Através de uma aferição de escalas, é permitido o estudo comparativo dos vários ficheiros introduzidos.

3.1. A partir do conjunto dos ficheiros aferidos, pode ser pedida a visualização das sucessivas configurações, assumidas em função dos crescentes valores do raio, expressos em quilómetros.²

É sempre de realçar o efeito da sobreposição dos círculos, sugestivo de uma ideia de intensidade de aglomeração.

3.2. A sequência dos grafos que surgem, para cada incremento do raio, quando dois ou mais círculos se intersectam, é outra opção gráfica possível para cada ficheiro.

3.3. Para o cálculo dos sucessivos valores da medida $C(N,r)$, é utilizado um algoritmo que faz a contagem, $n(r)$, das manchas, até ao caso limite: a obtenção de uma mancha única.

É possível visualizar a tabela de cálculo de $C(N,r)$.

3.4. O gráfico em escada da função $C(N,r)$ é também executado, decorrendo automaticamente dos valores anteriormente calculados.

Nota:

Importa referir que a construção dos grafos, bem como a contagem $n(r)$, não depende da qualidade gráfica das configurações das manchas, mas sim das semi-distâncias entre os pontos.

¹ Os autores preparam uma versão em linguagem *Basic* da *Microsoft* para *Apple Macintosh*, prontificando-se a fornecê-la.

² A expressão dos valores do raio noutra unidade é facilmente conseguida por adequada modificação nas instruções das linhas n.º 1960 e 2670 do programa.

LISTAGEM DO PROGRAMA *

```

100 REM *****
110 REM *****
120 KEY OFF
130 OPTION BASE 1
140 DIM T(1000,2): DIM B1$(10,12): DIM X(51): DIM Y(51)
150 DIM X2(51): DIM Y2(51): DIM M(51)
160 SCREEN 2
170 REM *** Menu principal ***
180 GOSUB 3070
190 LOCATE 4,31: PRINT "– MENU PRINCIPAL –"
200 LOCATE 8,15: PRINT "1 – Introduzir novos valores em ficheiro"
210 LOCATE 11,15: PRINT "2 – Operar com os ficheiros existentes"
220 LOCATE 14,15: PRINT "3 – Visualizar ficheiros existentes"
230 LOCATE 17,15: PRINT "0 – Abandonar o programa "
240 LOCATE 20,55: INPUT "Qual a opção "; OPÇÃO
250 IF OPÇÃO=1 THEN 300
260 IF OPÇÃO=2 THEN 1120
270 IF OPÇÃO=3 THEN 2760
280 IF OPÇÃO=0 THEN CLS: SYSTEM
290 GOTO 180
300 REM *** Introdução de valores ***
310 CLEAR
320 CLS: GOSUB 3070
330 LOCATE 3,20: PRINT "INTRODUÇÃO DE NOVOS VALORES EM FICHEIRO"
340 LOCATE 18,5: PRINT " < NOME > . fic "
350 LOCATE 19,5: PRINT " Número máximo de pontos – 50 "
360 LOCATE 7,20: INPUT "Qual o nome do novo ficheiro ";A1$
370 LOCATE 10,20: INPUT "Qual a escala da carta ( 1 / ... ) ";E1
380 LOCATE 13,20: INPUT "Qual o número de pontos ";N1
390 LOCATE 22,40: INPUT "Quer corrigir algum valor ( s / n ) ";IS
400 IF IS="s" OR IS="S" THEN GOTO 320
410 IF IS="n" OR IS="N" THEN GOTO 430
420 LOCATE 22,40: PRINT "                                     ": GOTO 390
430 DIM A1(N1+1,2): DIM A2(N1+1,2)
440 LET A1(1,1)=N1: LET A1(1,2)=E1/1000000!
450 FOR I=2 TO N1+1
460 GOSUB 3070
470 LOCATE 3,26: PRINT "INTRODUÇÃO DE NOVOS VALORES"
480 LOCATE 5,(80-LEN(A1$))/2: PRINT A1$
490 LOCATE 8,37: PRINT "Ponto"
500 LOCATE 10,38: PRINT I-1

```

(*) Sempre que aparecer o símbolo \div deve ser entendido como sinal de Cardinal.

```

510 LOCATE 15,16: PRINT "Coordenada x"
520 LOCATE 15,29: INPUT A1(I,1)
530 LOCATE 15,46: PRINT "Coordenada y"
540 LOCATE 15,59: INPUT A1(I,2)
550 NEXT I
560 LOCATE 17,34: PRINT "ULTIMO PONTO"
570 LOCATE 22,26: INPUT "Quer verificar os valores introduzidos ( s / n ) "; IS
580 IF IS="s" OR IS="S" THEN GOTO 610
590 IF IS="n" OR IS="N" THEN GOTO 970
600 LOCATE 22,26: PRINT "
: GOTO 570
610 REM *** Visualização dos valores introduzidos ***
620 GOSUB 3070
630 LOCATE 3,30: PRINT "VALORES INTRODUIZIDOS"
640 LOCATE 5,(80-LEN(A1$))/2: PRINT A1$
650 LOCATE 8,37: PRINT "Ponto"
660 LOCATE 15,16: PRINT "Coordenada x:"
670 LOCATE 15,46: PRINT "Coordenada y:"
680 FOR I=2 TO N1+1
690 LOCATE 10,38: PRINT I-1
700 LOCATE 15,29: PRINT A1(I,1)
710 LOCATE 15,59: PRINT A1(I,2)
720 LOCATE 22,48: PRINT "
730 LOCATE 22,48: PRINT "Prima uma tecla para continuar"
740 IF INKEY$="" THEN GOTO 740 ELSE GOTO 750
750 NEXT I
760 LOCATE 17,34: PRINT "ULTIMO PONTO"
770 LOCATE 22,40: PRINT "
780 LOCATE 22,40: INPUT "Quer corrigir algum valor ( s / n ) "; IS
790 IF IS="s" OR IS="S" THEN GOTO 820
800 IF IS="n" OR IS="N" THEN GOTO 970
810 LOCATE 22,40: PRINT "
": GOTO 780
820 REM *** Correção dos valores introduzidos ***
830 GOSUB 3070
840 LOCATE 3,30: PRINT "CORRECÇÃO DE VALORES"
850 LOCATE 5,(80-LEN(A1$))/2: PRINT A1$
860 LOCATE 8,27: INPUT "Qual o ponto a corrigir "; I
870 LOCATE 12,31: PRINT "Coordenadas actuais"
880 LOCATE 14,16: PRINT "Coordenada x: "; A1(I+1,1)
890 LOCATE 14,46: PRINT "Coordenada y: "; A1(I+1,2)
900 LOCATE 17,31: PRINT "Novas coordenadas"
910 LOCATE 19,16: INPUT "Coordenada x "; A1(I+1,1)
920 LOCATE 19,46: INPUT "Coordenada y "; A1(I+1,2)
930 LOCATE 22,40: INPUT "Quer continuar a correcção ( s / n ) "; IS
940 IF IS="s" OR IS="S" THEN GOTO 830
950 IF IS="n" OR IS="N" THEN GOTO 970
960 LOCATE 22,39: PRINT "
": GOTO 930
970 REM *** Gravação dos valores introduzidos ***

```



```

980 GOSUB 3070
990 LOCATE 11,14: PRINT "Prima < ENTER > para gravar os valores introduzidos"
1000 IF INKEY$="" THEN GOTO 1000 ELSE GOTO 1010
1010 OPEN "R", :+1,A1$,14
1020 FIELD :+1, 7 AS C1$, 7 AS C2$
1030 REM ***** RSET C1$=STR$(A1(1,1))
1040 REM ***** PUT :+1,1
1050 FOR I=1 TO N1+1
1060 RSET C1$=STR$(A1(I,1))
1070 RSET C2$=STR$(A1(I,2))
1080 PUT :+1,1
1090 NEXT I
1100 CLOSE :+1
1110 RUN
1120 REM *** Quais os ficheiros para operar ? ***
1130 GOSUB 3070
1140 LOCATE 3,23: PRINT "OPERAR COM OS FICHEIROS EXISTENTES"
1150 LOCATE 6,5: INPUT "Qual o número de ficheiros com que pretende operar "; M1
1160 FOR J=1 TO 10:FOR Y=1 TO 12 :B1$(J,Y)="" :NEXT Y:NEXT J
1170 FOR J=1 TO M1
1180 LOCATE J+8,5: INPUT "Nome do ficheiro "; B1$(J,12)
1190 NEXT J
1200 REM *** Determinação do factor de escala ***
1210 CLS
1220 XMI=10000
1230 XMA=0
1240 YMI=10000
1250 YMA=0
1260 FOR J=1 TO M1
1270 OPEN "R", :+1,B1$(J,12),14
1280 FIELD :+1, 7 AS C1$, 7 AS C2$
1290 GET :+1,1
1300 N1=VAL(C1$)
1310 FOR I=2 TO N1+1
1320 GET :+1,I
1330 X(I)=VAL(C1$)
1340 Y(I)=VAL(C2$)/2.4
1350 IF X(I) < XMI THEN XMI=X(I)
1360 IF X(I) > XMA THEN XMA=X(I)
1370 IF Y(I) < YMI THEN YMI=Y(I)
1380 IF Y(I) > YMA THEN YMA=Y(I)
1390 NEXT I
1400 CLOSE :+1
1410 NEXT J
1420 ESCX=592/(XMA-XMI)
1430 ESCY=144/(YMA-YMI)
1440 IF ESCX<=ESCY THEN ESC=ESCX
1450 IF ESCX>ESCY THEN ESC=ESCY

```

```

1460 REM *** Ficheiros dos quais pode gerar gráficos ***
1470 GOSUB 3070
1480 LOCATE 3,20: PRINT "FICHEIROS DOS QUAIS PODE GERAR GRÁFICOS"
1490 LOCATE 5,32: PRINT "Escalas aferidas"
1500 FOR J=1 TO M1
1510 LOCATE J+8,34: PRINT B1$(J, 12),
1520 NEXT J
1530 LOCATE 22,48: PRINT "Prima uma tecla para continuar"
1540 IF INKEY$="" THEN GOTO 1540 ELSE GOTO 1550
1550 GOSUB 3070
1560 LOCATE 11,8: INPUT "Qual o nome do ficheiro de que pretende gerar gráficos"; D1$
1570 OPEN "R", :+1, D1$, 14
1580 FIELD :+1, 7 AS C1$, 7 AS C2$
1590 GET :+1, 1
1600 N2=VAL(C1$): ESCM=VAL(C2$)
1610 FOR I=2 TO N2+1
1620 GET :+1,I
1630 X2(I)=VAL(C1$)
1640 Y2(I)=VAL(C2$)/2.4
1650 NEXT I
1660 CLOSE :+1
1670 REM *** Menu - gráficos ***
1680 GOSUB 3070
1690 LOCATE 4,30: PRINT "- MENU - GRÁFICOS -"
1700 LOCATE 9,15: PRINT "1 - Configuração das manchas"
1710 LOCATE 11,15: PRINT "2 - Grafos das ligações entre as manchas"
1720 LOCATE 13,15: PRINT "3 - Função C ( N,r )"
1730 LOCATE 15,15: PRINT "4 - Outros ficheiros"
1740 LOCATE 17,15: PRINT "0 - Menu principal"
1750 LOCATE 20,55: INPUT "Qual a opção "; OPÇÃO
1760 IF OPÇÃO=1 THEN 1820
1770 IF OPÇÃO=2 THEN 2420
1780 IF OPÇÃO=3 THEN 2050
1790 IF OPÇÃO=4 THEN 1470
1800 IF OPÇÃO=0 THEN 180
1810 GOTO 1680
1820 REM *** configuração das manchas ***
1830 GOSUB 3070
1840 LOCATE 11,15: INPUT "Qual o título";D2$
1850 CLS
1860 LINE (0,0)-(639,191),,B
1870 LINE (2,2)-(637,189),,B
1880 GOSUB 2860
1890 RAIO=1: NL=1
1900 FOR J=2 TO N2+1
1910 XR=INT(((X2(J)-XMI)*ESC)+1)+23
1920 YR=191-INT(((Y2(J)-YMI)*ESC)+1)-23/2.4
1930 CIRCLE (XR,YR),RAIO

```

```

1940 NEXT J
1950 LOCATE 2,2: PRINT "
1960 LOCATE 2,2: PRINT "Raio (Km) = "; INT(RAIO/ESC * ESCM*100)/100;
1970 LOCATE 2,20+(60-LEN(D2$))/2: PRINT D2$
1980 LOCATE 25,1: PRINT "
1990 LOCATE 25,1: INPUT "Aumentar raio ( A ); Menu - gráficos ( M ) "; SS
2000 IF SS="a" OR SS="A" THEN GOTO 2030
2010 IF SS="m" OR SS="M" THEN GOTO 1680
2020 LOCATE 25,1: PRINT "
2030 RAIO=RAIO+3
2040 GOTO 1900
2050 REM *** Função C(N,r) ***
2060 GOSUB 3070
2070 LOCATE 9,20: PRINT "UM MOMENTO ...": LOCATE 13,36: PRINT "ESTOU A
FAZER OS CÁLCULOS"
2080 RAIO=1: NL=0: NM=0
2090 WHILE NM<>1
2100 FOR J=2 TO N2+1
2110 XR=INT(((X2(J)-XMI)*ESC)+1)+23
2120 YR=191-INT(((Y2(J)-YMI)*ESC)+1)-23/2.4
2130 NEXT J
2150 GOSUB 3160
2160 IF NM=1 THEN GOTO 2200
2170 RAIO=RAIO+3
2180 GOTO 2100
2190 WEND
2200 GOSUB 3500
2210 LOCATE 2,10+(70-LEN(D2$))/2: PRINT D2$
2220 LOCATE 25,1: PRINT "
2230 LOCATE 25,1: INPUT "Visualizar valores ( V ); Menu - gráficos ( M ) "; SS
2240 IF SS="v" OR SS="V" THEN GOTO 2270
2250 IF SS="m" OR SS="M" THEN GOTO 1680
2260 GOTO 2220
2270 REM *** Visualização de valores ****
2280 GOSUB 3070
2290 LOCATE 3,31: PRINT «Cálculo da função»
2300 LOCATE 5,22: PRINT "r": LOCATE 5,32: PRINT "n ( r)": LOCATE 5,43:
PRINT "C ( N,r )
2310 LOCATE 7,21: PRINT "0": LOCATE 7,35: PRINT N2: LOCATE 7,47: PRINT "0"
2320 W=8
2330 H=N2: FOR K=1 TO NL
2340 IF W=22 THEN GOSUB 3410
2350 IF T(K,2)<H THEN LOCATE W,20: PRINT INT(T(K,1)*100)/100: LOCATE W,35:
PRINT T(K,2): LOCATE W,46: PRINT (N2-T(K,2))/(N2-1): H=T(K,2): W=W+1
2360 NEXT K
2370 LOCATE 25,1: PRINT "
2380 LOCATE 25,1: INPUT "Gráfico ( G ); Menu - gráficos ( M ) ";SS
2390 IF SS="g" OR SS="G" THEN GOTO 2200

```



```

2400 IF S$="m" OR S$="M" THEN GOTO 1680
2410 GOTO 2370
2420 REM *** Grafos ***
2430 RAIO=1: RRAIO =RAIO/ESC*2
2440 GOSUB 3070
2450 LOCATE 11,15: INPUT "Qual o título";D2$
2460 CLS: REM ***** LET ESCM=.25
2470 LINE (0,0)-(639,191),,B
2480 LINE (2,2)-(637,189),,B
2490 GOSUB 2860
2500 FOR J=2 TO N2+1
2510 XR=INT(((X2(J)-XMI)*ESC)+1)+23
2520 YR=191-INT(((Y2(J)-YMI)*ESC)+1)-23/2.4
2530 CIRCLE (XR,YR), 1
2540 NEXT J
2550 R=1
2560 FOR J=2 TO N2
2570 R=R+1
2580 FOR JJ=R TO N2+1
2590 DIS=SQR(((X2(J)-X2(JJ))^2+((Y2(J)-Y2(JJ))^2.4)^2)
2600 A=INT((X2(J)-XMI)*ESC+1)+23
2610 B=191-INT((Y2(J)-YMI)*ESC+1)-23/2.4
2620 C=INT((X2(JJ)-XMI)*ESC+1)+23
2630 D=191-INT((Y2(JJ)-YMI)*ESC+1)-23/2.4
2640 IF DIS<=RRAIO THEN LINE (A,B)-(C,D)
2650 NEXT JJ
2660 NEXT J
2670 LOCATE 2,2: PRINT "Raio (Km) = "; INT(RAIO/ESC * ESCM*100)/100;
2680 LOCATE 2,20+(60-LEN(D2$))/2: PRINT D2$
2690 LOCATE 25,1: PRINT ""
2700 LOCATE 25,1: INPUT "Aumentar raio ( A ); Menu - gráficos ( M )"; S$
2710 IF S$="a" OR S$="A" THEN GOTO 2740
2720 IF S$="m" OR S$="M" THEN GOTO 1680
2730 GOTO 2690
2740 RAIO=RAIO+3: RRAIO=RAIO/ESC*2
2750 GOTO 2460
2760 REM *** Ficheiros existentes ***
2770 CLS
2780 LOCATE 1,1: PRINT "*****"
*****
2790 LOCATE 23,1: PRINT "*****"
*****
2800 LOCATE 3,30: PRINT "FICHEIROS EXISTENTES"
2810 FILES "*.FIC"
2820 LOCATE 25,1: PRINT ""
2830 LOCATE 25,1: INPUT "Menu ( M )"; S$
2840 IF S$="m" OR S$="M" THEN GOTO 180
2850 GOTO 2820

```

```

2860 REM *** Escala ***
2870 IF (XMA - XMI) >= (YMA - YMI) THEN CAI=XMA - XMI
2880 IF (XMA - XMI) < (YMA - YMI) THEN CAI=YMA - YMI
2890 LET CAI=CAI/8*ESCM
2900 IF INT(CAI) <= 5 THEN CC=5
2910 IF INT((CAI > 5)) AND INT ((CAI <= 10)) THEN CC=10
2920 IF INT((CAI > 10)) AND INT((CAI <= 25)) THEN CC=25
2930 IF INT((CAI >25)) AND INT((CAI <= 50)) THEN CC=50
2940 IF INT((CAI >50)) AND INT((CAI <= 75)) THEN CC=75
2950 IF INT((CAI >75)) AND INT((CAI <= 100)) THEN CC=100
2960 IF INT((CAI >100)) AND INT((CAI <=150)) THEN CC=150
2970 IF INT((CAI >150)) AND INT((CAI <= 200)) THEN CC=200
2980 LET C=CC*ESC/ESCM
2990 LINE (600,182) - (600 - C,182)
3000 LINE (600,180) - (600 - C,180)
3010 LINE (600,181) - (600 - C/3,181)
3020 LINE (600 - 2/3*C,181) - (600 - C,181)
3030 LOCATE 22,(600 - C)/8: PRINT 0
3040 LOCATE 22,74: PRINT CC;"Km"
3050 RETURN
3060 REM *** Moldura do quadro ***
3070 CLS
3080 LOCATE 1,1: PRINT "*****"
*****
3090 FOR B=1 TO 80 STEP 79
3100 FOR A=2 TO 22
3110 LOCATE A,B: PRINT "*"
3120 NEXT A
3130 NEXT B
3140 LOCATE 23,1: PRINT "*****"
*****
3150 RETURN
3160 REM *** Contagem de manchas ***
3170 NL=NL+1
3180 FOR II=2 TO N2+1 : M(II)=0 : NEXT II
3190 RRAIO =RAIO/ESC*2
3200 NM=0:K=2
3210 WHILE K<N2+2
3220 NM =NM+1:OK=1
3230 II=2
3240 WHILE M(II) <> 0
3250 II=II+1
3260 WEND
3270 M(II)=NM
3280 WHILE OK>0
3290 II=2
3300 WHILE M(II) < -NM
3310 II=II+1

```

```

3320 WEND
3330 OK=OK-1:M(II)=-M(II):K=K+1
3340 FOR JJ=2 TO N2+1
3350 IF M(JJ)=0 AND (SQR(((X2(II)-X2(JJ))^2)+((Y2(II)-Y2(JJ))*2.4)^2))<=
  RRAIO THEN OK=OK+1:M(JJ)=NM
3360 NEXT JJ
3370 WEND
3380 WEND
3390 T(NL,1)=RRAIO/2*ESCM:T(NL,2)=NM
3400 RETURN
3410 REM *** formato na visualização ***
3420 LOCATE 25,2:PRINT "Prima uma tecla para continuar";IS
3430 IF INKEY$="" THEN GOTO 3430 ELSE GOTO 3440
3440 W=7
3450 CLS
3460 GOSUB 3070
3470 LOCATE 3,31:PRINT "Cálculo da função"
3480 LOCATE 5,22:PRINT "r":LOCATE 5,32:PRINT "n ( r )":LOCATE 5,43:
  PRINT "C ( N,r )"
3490 RETURN
3500 REM *** Gráfico de saída C (N,r) ***
3510 CLS
3520 GOSUB 3070
3530 LOCATE 11,15:INPUT "Qual o título";D2$
3540 CLS
3550 LINE (0,0)-(639,191),,B
3560 LINE (2,2)-(637,189),,B
3570 IF T(NL,1)<200 THEN DELTAX=20
3580 IF T(NL,1)<150 THEN DELTAX=15
3590 IF T(NL,1)<100 THEN DELTAX=10
3600 IF T(NL,1)<50 THEN DELTAX=5
3610 IF T(NL,1)<20 THEN DELTAX=2
3620 IF T(NL,1)<10 THEN DELTAX=1
3630 IF T(NL,1)<5 THEN DELTAX=.5
3640 FACX=500/(DELTAX*10)
3650 LINE (63,136)-(576,136)
3660 LINE (63,136)-(63,26)
3670 FOR VX=1 TO 10
3680 LINE (63+VX*DELTAX*FACX,134)-(63+VX*DELTAX*FACX,138)
3690 LOCATE 19,INT((63+VX*DELTAX*FACX)/8-1):PRINT VX*DELTAX
3700 NEXT VX
3710 FOR VY=1 TO 10
3720 LINE(61,136-11*VY)-(65,136-11*VY)
3730 NEXT VY
3740 LOCATE 4,4:PRINT "1.0":LOCATE 11,4:PRINT "0.5":LOCATE 18,4:PRINT
  "0.0":LOCATE 2,3:
3750 LOCATE 2,3:PRINT "C(N,r)":LOCATE 22,37:PRINT "r (Km)"
3760 FOR I=1 TO NL-1

```

3770 LINE (63+T(I,1)*FACX,136-110*((N2-T(I,2))/(N2-1)))-(63+T(I+1,1)*FACX
136-110*((N2-T(I,2))/(N2-1)))
3780 NEXT I
3790 LINE(63+T(NL,1)*FACX,136-110)-(576,136-110)
3800 RETURN

REFERÊNCIA BIBLIOGRÁFICA

KOSTRUBIEC, B. e SALGUEIRO, G. (1986) — «Medida de Concentração Espacial e Relativa
Duma Distribuição Pontual (Ensaio na Região Centro de Portugal)»
IV Colóquio Ibérico de Geografia, Actas, Coimbra, I. E. G., p. 845-855.